Electronic Supplementary Material (ESI) for Physical Chemistry Chemical Physics.

Two-scale perturbation expansion

This file is Supplementary Information for the article *Pattern transitions in a compressible floating elastic sheet*, by Oz Oshri and Haim Diamant. It presents the detailed two-scale expansion of the elastic energy *G* to 4th order in $\epsilon = \sqrt{\mathrm{Pw} - P}$.

Initiation

```
P = Pw - ϵ^2;
```

The approximate height function:

```
h[s_] = ϵ Cos[k s] H[ϵ s];
```

Compression field

```
γ[s_] = γ0 + ϵ^2 γ2[s] + ϵ^4 γ4[s];
```

dϕ/ds is related to h(s) and γ(s) via the geometrical constraint sin(ϕ) = (1/γ) dh/ds [Eq. 1 in the article].

```
ϕdot[s_] = ∂s (h'[s] / γ[s]) / Sqrt[1 - (h'[s] / γ[s])^2];
```

We will also need later the function cos(ϕ).

```
cosϕ[s_] = Sqrt[1 - (h'[s] / γ[s])^2];
```

After taking the derivatives we replace the argument of the H's by S, to avoid spurious terms in the later expansion in ϵ.

```
ϕdot[s_] = ϕdot[s] //. {s ϵ → S};
cosϕ[s_] = cosϕ[s] //. {s ϵ → S};
```

We also need the value of the compression at the boundary (Γ here is \bar{\gamma} in the article).

```
Γ = Γ0 + ϵ^2 Γ2 + ϵ^4 Γ4;
```

At equilibrium there is a relation between γ, Γ, and dϕ/ds [Eq. 12 in the article]

```
γrhs = Sqrt[Γ^2 - ξ ϕdot[s]^2];
```

Expand this expression. First identify the 0th order compression.

```
γ0 = FullSimplify[γrhs /. {ϵ → 0}, Γ0 > 0]
```

$$\Gamma 0$$

and then the 2nd and 4th order terms,

```
γrhsexp = Series[γrhs, {ϵ, 0, 4}];
γ2[s_] = FullSimplify[Coefficient[γrhsexp, ϵ^2], Γ0 > 0]
```

$$\Gamma 2 - \frac{k^4\, \xi\, \mathrm{Cos}[k\, s]^2\, H[S]^2}{2\, \Gamma 0^3}$$

```
γ4[s_] = FullSimplify[Coefficient[γrhsexp, ϵ^4], Γ0 > 0]
```

$$\frac{1}{8\,\Gamma0^7}\left(8\,\Gamma0^7\,\Gamma4 + k^2\,\zeta\left(12\,k^2\,\Gamma0^3\,\Gamma2\,\mathrm{Cos}[k\,s]^2\,H[S]^2 + k^4\,H[S]^4\left(-5\,k^2\,\zeta\,\mathrm{Cos}[k\,s]^4 - \left(\Gamma0^2 - 2\,k^2\,\zeta\right)\mathrm{Sin}[2\,k\,s]^2\right) - 16\,\Gamma0^4\,\mathrm{Sin}[k\,s]^2\,H'[S]^2 + 8\,\Gamma0^4\,\mathrm{Cos}[k\,s]^2\,H[S]\,H''[S]\right)\right)$$

Stretching energy

$$\mathrm{Gs} = \frac{1}{2\,\zeta}\left(\gamma[s] - 1\right)^2;$$

Expand and identify the terms.

```
Gs0 = Gs /. ϵ → 0
```

$$\frac{(-1 + \Gamma0)^2}{2\,\zeta}$$

```
Gsexp = Series[Gs, {ϵ, 0, 4}];
```

```
Gs2 = FullSimplify[Coefficient[Gsexp, ϵ^2]]
```

$$\frac{(-1 + \Gamma0)\left(\Gamma2 - \frac{k^4\,\zeta\,\mathrm{Cos}[k\,s]^2\,H[S]^2}{2\,\Gamma0^3}\right)}{\zeta}$$

```
Gs4 = FullSimplify[Coefficient[Gsexp, ϵ^4]]
```

$$\frac{1}{8\,\Gamma0^7\,\zeta}\left(4\,\Gamma0^7\left(\Gamma2^2 + 2\,(-1 + \Gamma0)\,\Gamma4\right) + k^2\,\zeta\left(4\,k^2\,\Gamma0^3\,(-3 + 2\,\Gamma0)\,\Gamma2\,\mathrm{Cos}[k\,s]^2\,H[S]^2 + k^4\,H[S]^4\left(k^2\,(5 - 4\,\Gamma0)\,\zeta\,\mathrm{Cos}[k\,s]^4 - (-1 + \Gamma0)\left(\Gamma0^2 - 2\,k^2\,\zeta\right)\mathrm{Sin}[2\,k\,s]^2\right) - 16\,(-1 + \Gamma0)\,\Gamma0^4\,\mathrm{Sin}[k\,s]^2\,H'[S]^2 + 8\,(-1 + \Gamma0)\,\Gamma0^4\,\mathrm{Cos}[k\,s]^2\,H[S]\,H''[S]\right)\right)$$

Average over the fast oscillations

$$\mathrm{Gsav2} = \mathrm{FullSimplify}\left[\frac{k}{2\,\pi}\int_{-\pi/k}^{\pi/k}\mathrm{TrigExpand}[\mathrm{Gs2}]\,ds\right]$$

$$\frac{1}{4}\,(-1 + \Gamma0)\left(\frac{4\,\Gamma2}{\zeta} - \frac{k^4\,H[S]^2}{\Gamma0^3}\right)$$

$$\mathrm{Gsav4} = \mathrm{FullSimplify}\left[\frac{k}{2\,\pi}\int_{-\pi/k}^{\pi/k}\mathrm{TrigExpand}[\mathrm{Gs4}]\,ds\right]$$

$$\frac{1}{64\,\Gamma0^7\,\zeta}\left(32\,\Gamma0^7\left(\Gamma2^2 + 2\,(-1 + \Gamma0)\,\Gamma4\right) + k^2\,\zeta\left(16\,k^2\,\Gamma0^3\,(-3 + 2\,\Gamma0)\,\Gamma2\,H[S]^2 + k^4\left(-4\,(-1 + \Gamma0)\,\Gamma0^2 + k^2\,(7 - 4\,\Gamma0)\,\zeta\right)H[S]^4 - 64\,(-1 + \Gamma0)\,\Gamma0^4\,H'[S]^2 + 32\,(-1 + \Gamma0)\,\Gamma0^4\,H[S]\,H''[S]\right)\right)$$

Bending energy

$$\mathrm{Gb} = \frac{1}{2}\,\phi\mathrm{dot}[s]^2;$$

Expand and identify the terms (there is no 0th order bending term).

```
Gbexp = Series[Gb, {ϵ, 0, 4}];
```

```
Gb2 = FullSimplify[Coefficient[Gbexp, ϵ^2]]
```

$$\frac{k^4 \operatorname{Cos}[k\, s]^2\, H[S]^2}{2\, \Gamma 0^2}$$

```
Gb4 = FullSimplify[Coefficient[Gbexp, ϵ^4]]
```

$$\frac{1}{8\, \Gamma 0^6} k^2 \left(-8\, k^2\, \Gamma 0^3\, \Gamma 2 \operatorname{Cos}[k\, s]^2\, H[S]^2 + H[S]^4 \left(4\, k^6\, \zeta \operatorname{Cos}[k\, s]^4 + k^4 \left(\Gamma 0^2 - 2\, k^2\, \zeta\right) \operatorname{Sin}[2\, k\, s]^2\right) + 16\, \Gamma 0^4 \operatorname{Sin}[k\, s]^2\, H'[S]^2 - 8\, \Gamma 0^4 \operatorname{Cos}[k\, s]^2\, H[S]\, H''[S]\right)$$

Average over the fast oscillations

$$\text{Gbav2} = \text{FullSimplify}\left[\frac{k}{2\,\pi} \int_{-\pi/k}^{\pi/k} \text{TrigExpand}[\text{Gb2}]\, ds\right]$$

$$\frac{k^4\, H[S]^2}{4\, \Gamma 0^2}$$

$$\text{Gbav4} = \text{FullSimplify}\left[\frac{k}{2\,\pi} \int_{-\pi/k}^{\pi/k} \text{TrigExpand}[\text{Gb4}]\, ds\right]$$

$$\frac{1}{16\, \Gamma 0^6} k^2 \left(-8\, k^2\, \Gamma 0^3\, \Gamma 2\, H[S]^2 + k^4 \left(\Gamma 0^2 + k^2\, \zeta\right) H[S]^4 + 16\, \Gamma 0^4\, H'[S]^2 - 8\, \Gamma 0^4\, H[S]\, H''[S]\right)$$

Fluid energy

$$\text{Gf} = \frac{1}{2} \gamma[s] \left(h[s]^2 \;//.\; \{s\, \epsilon \to S\}\right) \cos\phi[s];$$

Expand and identify the terms (there is no 0th order fluid term)

```
Gfexp = Series[Gf, {ϵ, 0, 4}];
```

```
Gf2 = FullSimplify[Coefficient[Gfexp, ϵ^2]]
```

$$\frac{1}{2} \Gamma 0 \operatorname{Cos}[k\, s]^2\, H[S]^2$$

```
Gf4 = FullSimplify[Coefficient[Gfexp, ϵ^4]]
```

$$-\frac{1}{4\, \Gamma 0^3} \operatorname{Cos}[k\, s]^2\, H[S]^2 \left(-2\, \Gamma 0^3\, \Gamma 2 + H[S]^2 \left(k^4\, \zeta \operatorname{Cos}[k\, s]^2 + k^2\, \Gamma 0^2 \operatorname{Sin}[k\, s]^2\right)\right)$$

Average over the fast oscillations

$$\text{Gfav2} = \text{FullSimplify}\left[\frac{k}{2\,\pi} \int_{-\pi/k}^{\pi/k} \text{TrigExpand}[\text{Gf2}]\, ds\right]$$

$$\frac{1}{4} \Gamma 0\, H[S]^2$$

$$\text{Gfav4} = \text{FullSimplify}\left[\frac{k}{2\,\pi} \int_{-\pi/k}^{\pi/k} \text{TrigExpand}[\text{Gf4}]\, ds\right]$$

$$\frac{1}{4} \Gamma 2\, H[S]^2 - \frac{k^2 \left(\Gamma 0^2 + 3\, k^2\, \zeta\right) H[S]^4}{32\, \Gamma 0^3}$$

Pressure-displacement (work) term

```
Gp = -P (1 - γ[s] cosϕ[s]);
```

Expand and identify the terms

```
Gp0 = Gp /. ϵ → 0
```

$$-\text{Pw}\,(1-\Gamma 0)$$

```
Gpexp = Series[Gp, {ϵ, 0, 4}];
```

```
Gp2 = FullSimplify[Coefficient[Gpexp, ϵ^2]]
```

$$\frac{1}{2\,\Gamma 0^3}\left(2\,\Gamma 0^3\,(1-\Gamma 0+\text{Pw}\,\Gamma 2)-k^2\,\text{Pw}\,H[S]^2\left(k^2\,\zeta\,\text{Cos}[k\,s]^2+\Gamma 0^2\,\text{Sin}[k\,s]^2\right)\right)$$

```
Gp4 = FullSimplify[Coefficient[Gpexp, ϵ^4]]
```

$$\frac{1}{8\,\Gamma 0^7}\Big(-8\,\Gamma 0^7\,(\Gamma 2-\text{Pw}\,\Gamma 4)+ 4\,k^2\,\Gamma 0^3\,H[S]^2\left(k^2\,(\Gamma 0+3\,\text{Pw}\,\Gamma 2)\,\zeta\,\text{Cos}[k\,s]^2+\Gamma 0^2\,(\Gamma 0+\text{Pw}\,\Gamma 2)\,\text{Sin}[k\,s]^2\right)+ \frac{1}{2}\,k^4\,\text{Pw}\,H[S]^4\left(-10\,k^4\,\zeta^2\,\text{Cos}[k\,s]^4-2\,\Gamma 0^4\,\text{Sin}[k\,s]^4+k^2\,\zeta\left(-3\,\Gamma 0^2+4\,k^2\,\zeta\right)\text{Sin}[2\,k\,s]^2\right)- 4\,\text{Pw}\,\Gamma 0^4\left(\Gamma 0^2\,\text{Cos}[k\,s]^2+4\,k^2\,\zeta\,\text{Sin}[k\,s]^2\right)H'[S]^2+8\,k^2\,\text{Pw}\,\Gamma 0^4\,\zeta\,\text{Cos}[k\,s]^2\,H[S]\,H''[S]\Big)$$

Average over the fast oscillations

```
Gpav2 = FullSimplify[k/(2 π) ∫_{-π/k}^{π/k} TrigExpand[Gp2] ds]
```

$$1-\Gamma 0+\text{Pw}\,\Gamma 2-\frac{k^2\,\text{Pw}\left(\Gamma 0^2+k^2\,\zeta\right)H[S]^2}{4\,\Gamma 0^3}$$

```
Gpav4 = FullSimplify[k/(2 π) ∫_{-π/k}^{π/k} TrigExpand[Gp4] ds]
```

$$-\frac{1}{64\,\Gamma 0^7}\Big(64\,\Gamma 0^7\,(\Gamma 2-\text{Pw}\,\Gamma 4)-16\,k^2\,\Gamma 0^3\left(\Gamma 0^2\,(\Gamma 0+\text{Pw}\,\Gamma 2)+k^2\,(\Gamma 0+3\,\text{Pw}\,\Gamma 2)\,\zeta\right)H[S]^2+ k^4\,\text{Pw}\left(3\,\Gamma 0^4+6\,k^2\,\Gamma 0^2\,\zeta+7\,k^4\,\zeta^2\right)H[S]^4+ 16\,\text{Pw}\,\Gamma 0^4\left(\Gamma 0^2+4\,k^2\,\zeta\right)H'[S]^2-32\,k^2\,\text{Pw}\,\Gamma 0^4\,\zeta\,H[S]\,H''[S]\Big)$$

Collecting the energy terms

0th order

```
G0 = FullSimplify[Gs0 + Gp0]
```

$$\frac{(-1+\Gamma 0)\,(-1+\Gamma 0+2\,\text{Pw}\,\zeta)}{2\,\zeta}$$

Minimize with respect to Γ0 and get the flat-to-wrinkle compression in terms of Pw

```
γwP = Γ0 /. Solve[∂_Γ0 G0 == 0, Γ0][[1]]
```

$$1-\text{Pw}\,\zeta$$

2nd order - flat-to-wrinkle transition

```
Gav2 = FullSimplify[(Gsav2 + Gbav2 + Gfav2 + Gpav2) /. Γ0 → γwP]
```

$$\frac{1}{4\,(-1+\mathrm{Pw}\,\zeta)^2}\left(4\,\mathrm{Pw}\,\zeta\,(-1+\mathrm{Pw}\,\zeta)^2+\left(k^4+k^2\,\mathrm{Pw}\,(-1+\mathrm{Pw}\,\zeta)-(-1+\mathrm{Pw}\,\zeta)^3\right)H[S]^2\right)$$

Going out of plane will decrease the energy when the coefficient of H^2 changes sign from positive to negative.

```
c2H2 = FullSimplify[(Coefficient[Gav2, H[S]^2]) //. {(-1 + Pw ζ) → -γw}]
```

$$\frac{k^4-k^2\,\mathrm{Pw}\,\gamma w+\gamma w^3}{4\,\gamma w^2}$$

Solve for Pw

```
Pwk[k_] = Pw /. Solve[c2H2 == 0, Pw][[1]]
```

$$\frac{k^4+\gamma w^3}{k^2\,\gamma w}$$

Find the critical k for which Pw is minimum

```
kγ = k /. Solve[Pwk'[k] == 0, k][[4]]
```

$$\gamma w^{3/4}$$

Get Pw as a function of γw

```
Pwγ = FullSimplify[Pwk[kγ]]
```

$$2\sqrt{\gamma w}$$

And in terms of ζ:

```
γwζ = FullSimplify[γw /. Solve[Pwγ == (1 - γw) / ζ, γw][[1]], ζ > 0]
```

$$1+2\,\zeta\left(\zeta-\sqrt{1+\zeta^2}\right)$$

```
Pwζ = FullSimplify[Pwγ /. γw → γwζ, ζ > 0]
```

$$2\sqrt{1+2\,\zeta\left(\zeta-\sqrt{1+\zeta^2}\right)}$$

```
Pwζ = 2 (Sqrt[1 + ζ^2] - ζ); (* Same but nicer *)
```

```
kζ = FullSimplify[kγ /. γw → γwζ, ζ > 0]
```

$$\left(1+2\,\zeta\left(\zeta-\sqrt{1+\zeta^2}\right)\right)^{3/4}$$

```
kζ = (-ζ + Sqrt[1 + ζ^2])^(3/2); (* Same but nicer *)
```

4th order

```
Gav4 =
 FullSimplify[(Gsav4 + Gbav4 + Gfav4 + Gpav4) /. {Γ0 → γwζ, Pw → Pwζ, k → kζ}, ζ > 0]
```

$$\frac{1}{64\,\zeta}\left(32\,\Gamma2\,(\Gamma2-2\,\zeta)+\zeta\left(16\sqrt{1+\zeta^2}\,H[S]^2+\left(\zeta-4\sqrt{1+\zeta^2}\right)H[S]^4+32\left(\zeta+\sqrt{1+\zeta^2}\right)H'[S]^2-32\left(\zeta+\sqrt{1+\zeta^2}\right)H[S]\,H''[S]\right)\right)$$

Minimize with respect to Γ2

```
Γ2ζ = Γ2 /. Solve[∂Γ2 Gav4 == 0, Γ2][[1]]
```

$$\zeta$$

```
Gav4 = FullSimplify[Gav4 /. Γ2 → Γ2ζ]
```

$$\frac{1}{64}\left(-32\,\zeta+16\sqrt{1+\zeta^2}\,H[S]^2+\left(\zeta-4\sqrt{1+\zeta^2}\right)H[S]^4+32\left(\zeta+\sqrt{1+\zeta^2}\right)H'[S]^2-32\left(\zeta+\sqrt{1+\zeta^2}\right)H[S]\,H''[S]\right)$$

Extract the coefficients of the four terms: H^2, H^4, H'^2, HH''

```
c4H2 = FullSimplify[Coefficient[Gav4, H[S]^2], ζ > 0]
```

$$\frac{\sqrt{1+\zeta^2}}{4}$$

```
c4H4 = FullSimplify[Coefficient[Gav4, H[S]^4], ζ > 0]
```

$$\frac{1}{64}\left(\zeta-4\sqrt{1+\zeta^2}\right)$$

```
c4Hd2 = FullSimplify[Coefficient[Gav4, H'[S]^2], ζ > 0]
```

$$\frac{1}{2}\left(\zeta+\sqrt{1+\zeta^2}\right)$$

```
c4HHdd = FullSimplify[Coefficient[Gav4, H[S] H''[S]], ζ > 0]
```

$$\frac{1}{2}\left(-\zeta-\sqrt{1+\zeta^2}\right)$$

The compression field (to order ϵ^2)

```
comp[s_] = FullSimplify[(γ0 + ϵ^2 γ2[s]) /. {S → ϵ s, Γ2 → Γ2ζ}]
```

$$\Gamma0+\epsilon^2\left(\zeta-\frac{k^4\,\zeta\,\mathrm{Cos}[k\,s]^2\,H[s\,\epsilon]^2}{2\,\Gamma0^3}\right)$$

The displacement (to order ϵ^2)

```
Δexp = Series[1 - γ[s] cosϕ[s], {ϵ, 0, 2}]
```

$$(1 - \Gamma 0) + \left(-\Gamma 2 + \frac{k^4\, \zeta\, \mathrm{Cos}[k\, s]^2\, \mathrm{H}[\mathrm{S}]^2}{2\, \Gamma 0^3} + \frac{k^2\, \mathrm{H}[\mathrm{S}]^2\, \mathrm{Sin}[k\, s]^2}{2\, \Gamma 0}\right) \epsilon^2 + \mathrm{O}[\epsilon]^3$$

```
Δ = FullSimplify[(L Integrate[Δexp, {s, -Pi / k, Pi / k}] / (2 Pi / k)) /.
    {Γ2 → Γ2ζ, k → kζ, Γ0 → γwζ}, ζ > 0]
```

$$\frac{1}{4}\, \mathrm{L} \left(-4\, \zeta \left(\epsilon^2 + 2\, \zeta - 2\sqrt{1 + \zeta^2}\right) + \epsilon^2 \sqrt{1 + \zeta^2}\; \mathrm{H}[\mathrm{S}]^2\right)$$